Appendix **B**

SAS Code for Multilevel Meta-Analysis

The meta-analytic dataset "tutoring.xlsx" is used. The dataset contains data from 28 studies with a total of 130 cases. The variables "Case" and "Study" refer to specific cases and studies respectively. The variable "D" refers to the immediate treatment effect and the variable "Dt" to the treatment effect on the time trend. One moderator variable is coded, namely "Age". The variable Age is mean-centered and coded as "Age_c" to avoid multicollinearity. The mean Age is 8.27 years. As a consequence Age_c = Age - 8.27.

In addition, the dataset contains the estimated effect size ("Estimate"), standardized effect size ("EstimateS") and bias-corrected standardized effect size ("Estimate_bc") for the immediate treatment effect (when D = 1 and Dt - 0) and the treatment on the time trend (when D = 0 and Dt = 1) per case (going from Case = 1 to Case = 130). The precision ("Precision"), Standardized precision ("Precision_bc") is also provided.

The following SAS code can be used to import the dataset "tutoring" and to conduct the multilevel meta-analysis.

• SAS Code for multilevel meta-analysis

```
/* Import Excel file "tutoring.xlsx */
```

RUN;

The "IMPORT" procedure is used to import the Excel data file ("tutoring.xlsx) in SAS. The command "PROC IMPORT OUT" assigns a name for the data file, for instance, tutoring. The next command, "DATAFILE=" specifies the exact location of the Excel data file on the computer. The "DBMS=xlsx REPLACE" command opens the Excel and overwrites any other file on the computer that exist with the same name. The statement SHEET = 'Tutoring" tells SAS to read the Excel sheet named "Tutoring" and "GETNAMES=YES" indicates that the variable names in the first row of the Excel file will be used as variable names in the SAS dataset.

/* Multilevel Meta-Analysis of Standardized Bias Corrected Effect Sizes*/

```
PROC MIXED DATA=tutoring COVTEST;
CLASS Study Case;
WEIGHT precision_bc;
MODEL estimate_bc = D DT/ NOINT SOLUTION DDFM=KENWARDROGER;
RANDOM D DT/ SUB=Study SOLUTION;
RANDOM D DT/ SUB=Case(Study) SOLUTION;
PARMS 1 1 1 1 1 / HOLD=5;
ODS OUTPUT solutionF=fixed covparms=random fitstatistics=fit
SolutionR=Ran;
```

RUN;

The "PROC MIXED" statement stands for an extension of the general linear model, which allows conducting a three-level meta-analysis. The "DATA" statement refers to the dataset to be used in the analysis (i.e., "tutoring"). The statement "COVTEST" generates "asymptotic standard errors and Wald Z-tests for the covariance parameter estimates" (SAS, 2018). Next, "CLASS" specifies the study level and case level (which are variables coded in the dataset). The next line includes the weight ("WEIGHT"). This is coded in the tutoring dataset as "precision bc".

The following line of code represents the "MODEL" statement. A regression equation is displayed with an estimated effect size (i.e., "estimate bc") that is equal to the immediate treatment effect (i.e., when D = 1 and Dt = 0 in the tutoring dataset) or the treatment effect on the time trend (i.e., when D = 0 and Dt = 1 in the tutoring dataset). "NOINT" suppresses the estimation of the intercept and the "SOLUTION" option requests the fixed effects parameter estimates. The "DDFM" statement specifies the method of denominator degrees of freedom, which is the "KENWARDROGER" approximation. The next line of code specifies the random effects: D and Dt are allowed to very between studies ("SUB = Study") and D and Dt are allowed to vary between cases within studies ["SUB = Case(Study)"]. The solution option is requesting a G matrix with the estimates of the random effect parameters. Because we are conducting a meta-analysis, the within variability at the first level is assumed to be known (was estimated using OLS and used as weight in the analysis) and therefore we keep this fixed at 1 using the "PARMS" statement. In total there are 5 random components: the between-study variance of the immediate treatment effect and the treatment effect on the time trend, the between-case variance of the immediate treatment effect and the treatment effect on the time trend and the residual within-case variance. Therefore, there are 5 PARMS values. The first 4 PARM values can get any starting value (here we assigned a 1) and the last PARM value is referring to the within-case residuals variance and is given a value of 1 and is kept fixed ("HOLD = 5"). Last, the "ODS OUTPUT" statement asks to save the fixed effects (solutionF=fixed), the random effects (covparms=random), the fit indices (fitstatistics=fit) and the case-specific and study specific estimates (SolutionR=Ran).

The following output table for the treatment effect estimates is obtained:	

Solution for Fixed Effects						
Effect	Estimate	Standard Error	DF	t Value	$Pr \geq [t]$	
D	3.7353	1.0851	22.6	3.44	0.0023	
Dt	0.09471	0.1934	24.4	0.49	0.6287	

The indicates that the overall average immediate treatment effect is estimated to be statistically significant and equals 3.74 [*SE* = 1.09, *t*(22.6) = 3.44, *p* =.0023]. The estimated treatment effect on the time trend equals 0.09 and is not statistically significant. [γ_{300} = 0.09, *SE* = 0.19, *t*(24.4) = 0.49, *p* =.63].

Covariance Parameter Estimates							
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr ≻ Z		
D	Study	26.7944	9.4659	2.83	0.0023		
Dt	Study	0.4955	0.2769	1.79	0.0368		
D	Case(Study)	18.6547	2.7149	6.87	<.0001		
Dt	Case(Study)	2.0667	0.3334	6.20	<.0001		
Residual		1.0000	0				

The following table for the random component effect estimates is obtained:

These results indicate that all the random components are estimated to be statistically significant.

In addition, the estimate of 28 study-specific and 130 case-specific treatment effect estimates are obtained. Because these results take up a lot of space, they are included in the supplementary materials.

/* Multilevel Meta-Analysis of Standardized Bias Corrected Effect Sizes with the Moderator "Age_c"*/

The following PROC mixed code can be used to include the second level predictor "Age_c". We mean-centered Age to avoid multicollinearity as an interaction between Age and D and Age and Dt was included. It is anticipated that Age_c will moderator the relation between the effectiveness of the treatment and the social outcome.

```
PROC MIXED DATA=tutoring COVTEST;
CLASS Study Case;
WEIGHT precision_bc;
MODEL estimate_bc = D DT D*Age_c Dt*Age_c/ NOINT SOLUTION
DDFM=KENWARDROGER;
RANDOM D DT/ SUB=Study SOLUTION;
RANDOM D DT/ SUB=Case(Study) SOLUTION;
PARMS 1 1 1 1 1/ HOLD=5;
ODS OUTPUT solutionF=fixed covparms=random fitstatistics=fit
SolutionR=Ran;
```

RUN;

The following output table for the treatment effect estimates is obtained:

Solution for Fixed Effects							
Effect	Estimate	Estimate Standard Error DF		t Value	Pr > [t]		
D	3.8811	1.0082	21.6	3.85	0.0009		
Dt	0.07843	0.1927	23.1	0.41	0.6878		
D*Age_c	-0.5765	0.2525	64.1	-2.28	0.0257		
Dt*Age_c	0.07875	0.05944	37.7	1.32	0.1932		

Based on these results, we found that Age_c moderators the relation between the immediate treatment effect and social outcomes [γ_{210} = -0.58, SE = 0.25, t(64.1) = -2.28, p = .0257]. The older the students, the less effective the peer-tutoring intervention. This is not the case for the interaction between Age_c and the treatment effect on the time trend outcomes [γ_{310} = 0.079, SE = 0.059, t(37.7) = 1.32, p = .1932]

The following table for the random component effect estimates is obtained:

Covariance Parameter Estimates							
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr ≻ Z		
D	Study	22.3725	8.3386	2.68	0.0036		
Dt	Study	0.4945	0.2830	1.75	0.0403		
D	Case(Study)	18.6519	2.7273	6.84	<.0001		
Dt	Case(Study)	2.0624	0.3332	6.19	<.0001		
Residual		1.0000	0				

These results indicate that the between-case variance in the immediate treatment effect estimates decreased from 26.79 (without Age_c) to 22.37 (with Age_c). The other sources of variance remained close to unchanged.